Let's say that you want to implement the following circuit in PSPICE:



R_1 is 1KOhm and D_1 and D_2 are identical diodes with the following PSPICE model:

.model diode d(Is=2.68n n=1.8)

1) To find the voltage transfer characteristic of the above limiter you need to define a DC input and perform a DC sweep on the input and look at the output voltage:

Create a new txt file and type in the following code:

```
Diode limiter
* the first line is ALWAYS assumed to be the title and if you define a
* component on the first line of your code it is ignored.

*circuit description. It is usually very helpful if you assign names (rather than numbers)
*to different nodes

R1 in out 1K
* you don't need to mention the unit. PSICE already knows that a resistor
* is being defined. If you are defining a capacitor of "one Farad" don't write 1F since
* PSPICE will assume that it's a 1fF capacitor
D1 out 0 diode
D2 0 out diode

.model diode d(Is=2.68n n=1.8)

Vi in 0 DC 0

.op
*to report the DC operating point in the output file
.DC Vi -5 5 0.01
*performing the DC sweep
.probe
*to open a data window and be able to look at different voltages vs Vi

.end
*anything written after this line is ignored
```

I. **Save the file as "filename.cir".**
II. **Click on "Simulation Queue" button on the left. PSPICE Simulation Manager window will open.**
III. **In this window filename.cir appears under "Simulation" (if you don't see the filename you may need to close PSPICE and rerun it).**
IV. **Click on filename.cir in the simulation manager window and click on "run".**
V. **"data display" window will open.**
VI. **In the data display window go to "add trace".**
VII. **You can choose V(out) (voltage at node "out"). You can also choose certain functions of the voltages at different nodes (for example you can plot 20*log10(V(out)/V(in)) if you are doing an AC sweep and want to draw the bode plot of the gain). If you choose V(out) you will see the following voltage transfer characteristic:**



2) To find the gain at zero DC bias vs frequency you can define an AC input and perform an AC sweep:

Diode limiter

R1 in out 1K
D1 out 0 diode
D2 0 out diode

.model diode d(Is=2.68n n=1.8)

Vi in 0 AC 1m

.op
.AC LIN 1000 10k 10meg

```
*performing the AC sweep
.probe
.end
```

Repeat steps I-VII from the previous section. If you choose V(out)/V(in) you will see the following plot:



3) There is also another way to find the gain at one particular frequency and that is to look at the output waveform when the input is a small-signal sinusoid. To do this you need to perform a transient simulation:

```
Diode limiter

R1 in out 1K
D1 out 0 diode
D2 0 out diode

.model diode d(Is=2.68n n=1.8)

Vi in 0 sin(0 1m 10k)

.op
.tran 1u 0.5m 0 1u
*performing the transient simulation
.probe
.end
```

Repeat steps I-VII from the previous section. If you choose V(out) you will see the following output voltage waveform: